Coriolis force

Masatsugu Sei Suzuki
Department of Physics, SUNY at Binghamton
(Date: July 22, 2013)

In classical mechanics, Newton's laws hold in all systems moving uniformly relative to each other (i.e., inertial frames) if they hold in one system. However, this is no longer valid if a system undergoes accelerations. It is required to find the new relations between the equations of motion in a fixed system and those in the accelerated system.

Here we consider the equation of motion for an object that is viewed from a rotating frame which rotates with the angular velocity Ω relative to an inertial frame. Here we are interested in a frame attached to the rotating earth. Although the angular velocity of the spinning earth is so small, the earth's rotation does have measurable effects on the motion of pendulum (known as Foucault pendulum), projectile, and other systems. As far as we know, Coriolis force related to the earth's rotation is not covered in typical standard textbooks of general physics. Since undergraduate students (taking the class of general physics) are very interested in these topics, here we will discuss the effect of the Coriolis force on the motions of systems on earth, as one of the example of the relative motion. We encounter a various kinds of differential equations to be solved. In order to avoid the complicated mathematics, we will use the Mathematica to solve these equations rigorously. We also us the numerical method for solving differential equation. We think that such ways might be helpful to students who do not well understand how to solve the second order differential equations.

Gaspard-Gustave de Coriolis or **Gustave Coriolis** (21 May 1792– 19 September 1843) was a French mathematician, mechanical engineer and scientist. He is best known for his work on the supplementary forces that are detected in a rotating frame of reference. Coriolis was the first to coin the term "work" for the transfer of energy by a force acting through a distance.

https://en.wikipedia.org/wiki/Gaspard-Gustave_Coriolis

Jean Bernard Léon Foucault (18 September 1819 – 11 February 1868) was a French physicist best known for the invention of the Foucault pendulum, a device demonstrating the effect of the Earth's rotation. He also made an early measurement of the speed of light, discovered eddy currents, and although he did not invent it, is credited with naming the gyroscope.

http://en.wikipedia.org/wiki/L%C3%A9on_Foucault

1. What is the Corilis force?

The Coriolis effect is a deflection of moving objects when they are viewed in a rotating reference frame. In a reference frame with clockwise rotation, the deflection is to the left of the motion of the object; in one with counter-clockwise rotation, the deflection is to the right. Although recognized previously by others, the mathematical expression for the Coriolis force appeared in an 1835 paper by French scientist Gaspard-Gustave Coriolis, in connection with the theory of water wheels. Early in the 20th century, the term *Coriolis force* began to be used in connection with meteorology.

Newton's laws of motion describe the motion of an object in a (non-accelerating) inertial frame of reference. When Newton's laws are transformed to a uniformly rotating frame of reference, the Coriolis and centrifugal forces appear. Both forces are proportional to the mass of the object.

(1) The Coriolis force is proportional to the rotation rate and the centrifugal force is proportional to its square. The Coriolis force acts in a direction perpendicular to the rotation axis and to the velocity of the body in the rotating frame and is proportional to the object's speed in the rotating frame.

(2) The centrifugal force acts outwards in the radial direction and is proportional to the distance of the body from the axis of the rotating frame. These additional forces are termed inertial forces, fictitious forces or *pseudo forces*.[1] They allow the application of Newton's laws to a rotating system. They are correction factors that do not exist in a non-accelerating or inertial reference frame.

https://en.wikipedia.org/wiki/Coriolis_effect

2. Rotation matrix (inertial and rotation reference frames)



Fig. Rotation of the coordinate axes. $\overrightarrow{OP} = \boldsymbol{r}_I = \boldsymbol{r}_R$. $\{\boldsymbol{e}_x, \boldsymbol{e}_y\}$; the old orthogonal basis. $\{\boldsymbol{e}_{Rx}, \boldsymbol{e}_{Ry,}\}$;. and the new orthogonal basis. The rotation angle is θ. The rotation axis is the *z* axis.

We assume that

$$\dot{\theta} = \omega , \qquad \theta = \omega t$$

with

$$\Omega = (0,0,\omega) .$$

The unit vectors in the fixed reference frame (Cartesian coordinate) is defined as

$$\boldsymbol{e}_x = (1,0,0), \qquad \boldsymbol{e}_y = (0,1,0), \qquad \boldsymbol{e}_z = (0,0,1)$$

The unit vectors of the rotation reference frame is defined as

$$\boldsymbol{e}_{Rx} = (\cos\theta, \sin\theta, 0), \qquad \boldsymbol{e}_{Ry} = (-\sin\theta, \cos\theta, 0), \qquad \boldsymbol{e}_{Rz} = (0,0,1)$$

We consider the position vector of

$$\boldsymbol{r}_I = x_I \boldsymbol{e}_x + y_I \boldsymbol{e}_y + z_I \boldsymbol{e}_z$$

in the fixed reference frame and

$$\boldsymbol{r}_R = x_R \boldsymbol{e}_{Rx} + y_R \boldsymbol{e}_{Ry} + z_R \boldsymbol{e}_{Rz}$$

in the rotating reference frame . Note that

$$\boldsymbol{r}_I = \boldsymbol{r}_R .$$

with

$$\begin{pmatrix} x_R \\ y_R \\ z_R \end{pmatrix} = \begin{pmatrix} \cos\theta & \sin\theta & 0 \\ -\sin\theta & \cos\theta & 0 \\ 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} x_I \\ y_I \\ z_I \end{pmatrix}$$

and

$$\begin{pmatrix} x_I \\ y_I \\ z_I \end{pmatrix} = \begin{pmatrix} \cos\theta & -\sin\theta & 0 \\ \sin\theta & \cos\theta & 0 \\ 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} x_R \\ y_R \\ z_R \end{pmatrix}$$

3. Mathematica

We now calculate the velocity $\dot{\boldsymbol{r}}_I$ and acceleration $\ddot{\boldsymbol{r}}_I$ by using Mathematica. The results are as follows.

```
Clear["Global`"];
eRx[t_] := {Cos[ω t], Sin[ω t], 0};
eRy[t_] := {-Sin[ω t], Cos[ω t], 0};
eRz[t_] := {0, 0, 1}; ω1 = {0, 0, ω};
rI1[t_] := rRx[t] eRx[t] + rRy[t] eRy[t] +
  rRz[t] eRz[t];

vI1 = D[rI1[t], t] // Simplify;
aI1 = D[rI1[t], {t, 2}] // FullSimplify;

{vI1.eRx[t], vI1.eRy[t], vI1.eRz[t]} //
 Simplify
```

$\{-\omega\, rRy[t] + rRx'[t],\ \omega\, rRx[t] + rRy'[t],\ rRz'[t]\}$

```
{Cross[ω1, rI1[t]].eRx[t],
  Cross[ω1, rI1[t]].eRy[t],
  Cross[ω1, rI1[t]].eRz[t]} // Simplify
```

$\{-\omega\, rRy[t],\ \omega\, rRx[t],\ 0\}$

```
{aI1.eRx[t], aI1.eRy[t], aI1.eRz[t]} //
 Simplify
```

$\{-\omega^2\, rRx[t] - 2\,\omega\, rRy'[t] + rRx''[t],$
$-\omega^2\, rRy[t] + 2\,\omega\, rRx'[t] + rRy''[t],\ rRz''[t]\}$

```
{ 2 m Cross[vI1, ω1].eRx[t],
  2 m Cross[vI1, ω1].eRy[t],
  2 m Cross[vI1, ω1].eRz[t]} // Simplify
```

$\{2\, m\, \omega\, (\omega\, rRx[t] + rRy'[t]),$
$2\, m\, \omega\, (\omega\, rRy[t] - rRx'[t]),\ 0\}$

```
KI1 = m Cross[ω1, Cross[ω1, rI1[t]]] // Simplify;

{KI1.eRx[t], KI1.eRy[t], KI1.eRz[t]} //
 Simplify
```

$\{-m\,\omega^2\, rRx[t],\ -m\,\omega^2\, rRy[t],\ 0\}$

4. The velocity:

The velocity in the rotation reference frame:

$$\dot{\boldsymbol{r}}_I \cdot \boldsymbol{e}_{Rx} = \dot{x}_R - \omega y_R ,$$

$$\dot{\boldsymbol{r}}_I \cdot \boldsymbol{e}_{Ry} = \dot{y}_R + \omega x_R ,$$

$$\dot{\boldsymbol{r}}_I \cdot \boldsymbol{e}_{Rz} = \dot{z}_R$$

from the result (the Mathematica calculation described above).

The vector $\boldsymbol{\omega} \times \boldsymbol{r}_I$

$$(\boldsymbol{\omega} \times \boldsymbol{r}_I) \cdot \boldsymbol{e}_{Rx} = -\omega y_R$$

$$(\boldsymbol{\omega} \times \boldsymbol{r}_I) \cdot \boldsymbol{e}_{Ry} = \omega x_R$$

$$(\boldsymbol{\omega} \times \boldsymbol{r}_I) \cdot \boldsymbol{e}_{Rz} = 0$$

Therefore we get

$$[\dot{\boldsymbol{r}}_I - (\boldsymbol{\omega} \times \boldsymbol{r}_I)] \cdot \boldsymbol{e}_{Rx} = \dot{x}_R = \boldsymbol{v}_R \cdot \boldsymbol{e}_{Rx}$$

$$[\dot{\boldsymbol{r}}_I - (\boldsymbol{\omega} \times \boldsymbol{r}_I)] \cdot \boldsymbol{e}_{Ry} = \dot{y}_R = \boldsymbol{v}_R \cdot \boldsymbol{e}_{Ry}$$

$$[\dot{\boldsymbol{r}}_I - (\boldsymbol{\omega} \times \boldsymbol{r}_I)] \cdot \boldsymbol{e}_{Rz} = \dot{z}_R = \boldsymbol{v}_R \cdot \boldsymbol{e}_{Rz}$$

We introduce the velocity vector $\boldsymbol{v}_R$, which is defined by

$$\begin{aligned} \boldsymbol{v}_R &= \dot{x}_R \boldsymbol{e}_{Rx} + \dot{y}_R \boldsymbol{e}_{Ry} + \dot{z}_R \boldsymbol{e}_{Rz} \\ &= \dot{\boldsymbol{r}}_I - (\boldsymbol{\Omega} \times \boldsymbol{r}_I) \\ &= \boldsymbol{v}_I - (\boldsymbol{\Omega} \times \boldsymbol{r}_I) \end{aligned}$$

or

$$\dot{\boldsymbol{r}}_I = \boldsymbol{v}_I = \boldsymbol{v}_R + \boldsymbol{\Omega} \times \boldsymbol{r}_I = \boldsymbol{v}_R + \boldsymbol{\Omega} \times \boldsymbol{r}_R$$

since

$$\boldsymbol{r}_I = \boldsymbol{r}_R$$

where $\boldsymbol{v}_R$ is the velocity of the particle relative to the rotating set of axes. ω is the constant angular velocity relative to the inertial system.

5. Acceleration

The acceleration in the rotation reference frame:

$$m\ddot{\boldsymbol{r}}_I \cdot \boldsymbol{e}_{Rx} = m\ddot{x}_R - 2m\omega\dot{y}_R - m\omega^2 x_R \,,$$

$$m\ddot{\boldsymbol{r}}_I \cdot \boldsymbol{e}_{Ry} = m\dot{y}_R + 2m\omega\dot{x}_R - m\omega^2 x_{2R} \,,$$

$$m\ddot{\boldsymbol{r}}_I \cdot \boldsymbol{e}_{Rz} = m\ddot{z}_R$$

from the result (the Mathematica calculation described above).

(a) The vector $2m(\dot{\boldsymbol{r}}_I \times \boldsymbol{\omega})$:

$$2m(\dot{\boldsymbol{r}}_I \times \boldsymbol{\omega}) \cdot \boldsymbol{e}_{Rx} = 2m\omega(\dot{y}_R + \omega x_R)$$

$$2m(\dot{\boldsymbol{r}}_I \times \boldsymbol{\omega}) \cdot \boldsymbol{e}_{Ry} = -2m\omega(\dot{x}_R - \omega y_R) \,.$$

$$2m(\dot{\boldsymbol{r}}_I \times \boldsymbol{\omega}) \cdot \boldsymbol{e}_{Rz} = 0$$

(b) The vector $m[\boldsymbol{\omega} \times (\boldsymbol{\omega} \times \boldsymbol{r}_I)]$:

$$m[\boldsymbol{\omega} \times (\boldsymbol{\omega} \times \boldsymbol{r}_I)] \cdot \boldsymbol{e}_{Rx} = -m\omega^2 x_R$$

$$m[\boldsymbol{\omega} \times (\boldsymbol{\omega} \times \boldsymbol{r}_I)] \cdot \boldsymbol{e}_{Ry} = -m\omega^2 y_R$$

$$m[\boldsymbol{\omega} \times (\boldsymbol{\omega} \times \boldsymbol{r}_I)] \cdot \boldsymbol{e}_{Rz} = 0$$

Then we have

$$[m\ddot{\boldsymbol{r}}_I + 2m(\dot{\boldsymbol{r}}_I \times \boldsymbol{\Omega}) + m[\boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_I)]] \cdot \boldsymbol{e}_{Rx} = (m\ddot{x}_R - 2m\omega\dot{y}_R - m\omega^2 x_R) + 2m\omega(\dot{y}_R + \omega x_R) - m\omega^2 x_R$$
$$= m\ddot{x}_R$$

$$[m\ddot{\boldsymbol{r}}_I + 2m(\dot{\boldsymbol{r}}_I \times \boldsymbol{\Omega}) + m[\boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_I)]] \cdot \boldsymbol{e}_{Ry} = (m\ddot{y}_R + 2m\omega\dot{x}_R - m\omega^2 y_R) - 2m\omega(\dot{x}_R - \omega y_R) - m\omega^2 y_R$$
$$= m\ddot{y}_R$$

$$\{m\ddot{\boldsymbol{r}}_I + 2m(\dot{\boldsymbol{r}}_I \times \boldsymbol{\Omega}) + m[\boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_I)]\} \cdot \boldsymbol{e}_{R_z} = m\ddot{z}_R$$

Then the acceleration of the particle relative to the rotating set of axes is obtained as

$$m\boldsymbol{a}_R = m\ddot{\boldsymbol{r}}_I + 2m(\dot{\boldsymbol{r}}_I \times \boldsymbol{\Omega}) + m\boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_I)$$

or

$$\boldsymbol{a}_R = \boldsymbol{a}_I + 2(\boldsymbol{v}_I \times \boldsymbol{\Omega}) + \boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_I)$$

where

$$\boldsymbol{a}_R = \ddot{x}_R \boldsymbol{e}_{Rx} + \ddot{y}_R \boldsymbol{e}_{Ry} + \ddot{z}_R \boldsymbol{e}_{Rz}$$

and

$$\boldsymbol{F}_I = m\boldsymbol{a}_I$$

is the net external force acting on the particle.

6. Newton's second law

The equation of motion which in the inertial system is simplily given by

$$\boldsymbol{F}_I = m\ddot{\boldsymbol{r}}_I = m\boldsymbol{a}_I$$

Then we get the effective force as

$$\boldsymbol{F}_R = m\boldsymbol{a}_R = \boldsymbol{F}_I + 2m(\boldsymbol{v}_I \times \boldsymbol{\Omega}) + m\boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_I) = \boldsymbol{F}_I - 2m(\boldsymbol{\Omega} \times \boldsymbol{v}_I) + m\boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_I)$$

or

$$
\begin{aligned}
\boldsymbol{F}_R &= \boldsymbol{F}_I + 2m(\dot{\boldsymbol{r}}_I \times \boldsymbol{\Omega}) + m\boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_I) \\
&= \boldsymbol{F}_I + 2m[\boldsymbol{v}_R + (\boldsymbol{\omega} \times \boldsymbol{r}_R)] \times \boldsymbol{\Omega} + m\boldsymbol{\omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_R) \\
&= \boldsymbol{F}_I + 2m(\boldsymbol{v}_R \times \boldsymbol{\Omega}) - 2m\boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_R) + m\boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_R) \\
&= \boldsymbol{F}_I - 2m(\boldsymbol{\Omega} \times \boldsymbol{v}_R) - m\boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_R)
\end{aligned}
$$

where

$$\boldsymbol{r}_I = \boldsymbol{r}_R ,$$

and

$$\dot{\boldsymbol{r}}_I = \boldsymbol{v}_I = \boldsymbol{v}_R + \boldsymbol{\Omega} \times \boldsymbol{r}_R .$$

7. Corioli force and centrifugal force

In the expression

$$
\begin{aligned}
\boldsymbol{F}_R &= m\boldsymbol{a}_R \\
&= \boldsymbol{F}_I + F_{cor} + F_{cf} \\
&= \boldsymbol{F}_I - 2m(\boldsymbol{\Omega} \times \boldsymbol{v}_R) - m\boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_R)
\end{aligned}
$$

where

$$\boldsymbol{F}_I = m\boldsymbol{a}_I ,$$

denotes the sum of all the forces as identified in any inertial frame,

$$\boldsymbol{F}_{cor} = -2m(\boldsymbol{\Omega} \times \boldsymbol{v}_R) ,$$

is called the Corioli force, and

$$\boldsymbol{F}_{cf} = -m\boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_R)$$

is called the centrifugal force.



Fig. The centrifugal force in the (x_R, y_R, z_R) space.

We note that

$$-2m(\boldsymbol{\omega} \times \boldsymbol{v}_R) = 2m\omega \dot{y}_R \boldsymbol{e}_{Rx} - 2m\omega \dot{x}_R \boldsymbol{e}_{Ry}$$

$$-m\boldsymbol{\omega} \times (\boldsymbol{\omega} \times \boldsymbol{v}_R) = m\omega^2 x_R \boldsymbol{e}_{Rx} + m\omega^2 y_R \boldsymbol{e}_{Ry}$$

8. Simple alternative explanation of the effect of Coriolis force

Suppose that turntable is rotating counter-clockwise with angular velocity Ω_0. At time t = 0, a ball is thrown away from the center to the edge of the turntable (radially outward). There is no contact between the ball and the truntable for simplicity.

(i) In the inertial frame of an observation on the ground, the net force on the ball is zero, and the ball follows a straight path. However, when the ball arrives at the edge of the turntable, the observer *A* has moved to the left (the position at A'). In other words, the ball misses the point A.



Fig. As seen in the non-rotating frame. The table rotates counter-clockwise.

(ii)- In the rotating frame, the observer A and the origin are in a line on a rotating turntable. At the t = 0, the ball is moving in the direction of line OA (radially outward). Because of the Coriolis force the ball veers to the right and misses the point A. The ball actually arrives at the point A'.



Fig. As seen in the rotating frame. Observation (denoted by blue points). The ball (denoted by red points). Because of the Coriolis force, the ball veers to the right and misses the point A. The ball actually arrives at the point A' on the edge of turntable.

Suppose that $\overline{OA} = v_0 t = v_0 t$ and the angle $\angle AOB = \Omega_0 t$ in the (x_R, y_R) plane. The position of the point A' at the time t is given by

$$x_R(t) = v_0 t \sin(\Omega_0 t), \qquad y_R(t) = v_0 t \cos(\Omega_0 t)$$

We make a plot of the position (x_R, y_R) as a function of t using Mathematica, where $v_0 = 1$ and $\Omega_0 = 1$.



Fig. ParametricPlot of ($x_R(t)$, $y_R(t)$), where $v_0 = 1$ and $\Omega_0 = 1$.

We note that

$$\ddot{x}_R = 2v_0\Omega_0 \cos(\Omega_0 t) - v_0 t\Omega_0^{\ 2} \sin(\Omega_0 t)$$

$$\ddot{y}_R = -v_0 t\Omega_0^{\ 2} \cos(\Omega_0 t) - 2v_0\Omega_0 \sin(\Omega_0 t)$$

and

$$\dot{x}_R = v_0 t\Omega_0 \cos(\Omega_0 t) + v_0 \sin(\Omega_0 t)$$

$$\dot{y}_R = v_0 \cos(\Omega_0 t) - v_0 t\Omega_0 \sin(\Omega_0 t)$$

9. The effect of the Centrifugal force on the gravity on the earth



Fig. The centrifugal force $\boldsymbol{F}_{cf} = m[(\boldsymbol{\Omega}\times\boldsymbol{r}_R)\times\boldsymbol{\Omega}] = m(\Omega_0{}^2 r_R \sin\lambda)\boldsymbol{e}_\rho$. $\boldsymbol{\Omega}\times\boldsymbol{r}_R = (\Omega_0 r_R \sin\lambda)\boldsymbol{e}_\phi$. The unit vector $\boldsymbol{e}_\rho$ which points radially outward from the axis of the rotation. The unit vector $\boldsymbol{e}_\phi$ which is directed tangentially along the circle with radius ρ.

$$\boldsymbol{F}_{cf} = m[(\boldsymbol{\Omega}\times\boldsymbol{r}_R)\times\boldsymbol{\Omega}] = m(\Omega_0\rho)\boldsymbol{e}_\phi\times\Omega_0\boldsymbol{e}_z = m(\Omega_0{}^2\rho)\boldsymbol{e}_\rho = m(\Omega_0{}^2 r_R\sin\lambda)\boldsymbol{e}_\rho.$$

$$\boldsymbol{v}_\phi = \boldsymbol{\Omega}\times\boldsymbol{r}_R = (\Omega_0 r_R\sin\lambda)\boldsymbol{e}_\phi = (\Omega_0\rho)\boldsymbol{e}_\phi$$

where

$$\rho = r_R\sin\lambda$$

If we momentarily let $v_\phi = \Omega \times \boldsymbol{r}_R = (\Omega_0 \rho)\boldsymbol{e}_\phi$, then the centrifugal force takes the familiar form, $\dfrac{m v_\phi^{\ 2}}{\rho}$.



Fig. The effective gravitational force which is sum of the original gravitational force $m\boldsymbol{g}_0$ and the centrifugal force $\boldsymbol{F}_{cf}$.

The effective gravitational force along the radial direction

$$g_{rad} = g_0 - \Omega_0^{\ 2} r_R \sin^2 \lambda = g_0 - \Omega_0^{\ 2} R_E \cos^2 \beta$$

The effective gravitational force along the tangential direction

$$g_{\tan} = \Omega_0^{\ 2} r_R \sin \lambda \cos \lambda = \Omega_0^{\ 2} R_E \sin \beta \cos \beta$$

where

$$\Omega_0^{\ 2} R_E = 0.0336983 \text{ m/s}^2 = 3.36983 \text{ cm/s}^2.$$

Since $g_0 = 9.8$ m/s^2, we see that, because of the centrifugal force, the value of g at the equator is about 0.3 % less than at the poles.

10. The magnitude of $\boldsymbol{F}_{cor}$ and $\boldsymbol{F}_{cf}$

$$F_{cor} = 2m\Omega_0 v_R \approx m\Omega_0 v_R\,, \qquad F_{cf} = m r_R \Omega_0^{\ 2}\,.$$

where v_R is the object's speed relative to the rotating frame of the earth. It is the speed as observed by us on the earth's surface. Then we have

$$\frac{F_{cor}}{F_{cf}} = \frac{\Omega_0 v_R}{\Omega_0^{\ 2} r_R} = \frac{v_R}{\Omega_0 r_R} \approx \frac{v_R}{\Omega_0 R_E} = \frac{v_R}{V}\,.$$

Here we assume that $r_R = R_E$ which is the radius of the earth. Using $R_E = 6.372 \times 10^6$ m, we get

$$V = 463.39 \text{ m/s} = 1036.59 \text{ miles/h}.$$

((**Note**))

When $v_R = V$, the centrifugal force F_{cf} is comparable to the Coriolis force F_{cor}. The gravitational force is slightly changed by the centriufugal force. Even if $v_R \ll V$ where F_{cf} is much smaller than F_{cor}, the Coriolis force can have appreciable effect (for example, with Foucault pendulum).

11. The case of $\boldsymbol{F}_I = 0$ (no external force)

Supposew that $\boldsymbol{F}_I = 0$. Then we have the differential equations for x_R and y_R in the rotatin frame,

$$\ddot{x}_R = 2\omega\dot{y}_R + \omega^2 x_R\,,$$

$$\ddot{y}_R = -2\omega\dot{x}_R + \omega^2 y_R\,,$$

with the initial condition; $x_R = 0$, $y_R = 0$, $\dot{x}_R = 0$, and $\dot{y}_R = v_0$, The solution of these differential equations are given by

$$x_R(t) = v_0 t\sin(\omega t)\,, \qquad y_R(t) = v_0 t\cos(\omega t)$$

for the rotation frame and

$$x_I(t) = 0\,, \qquad y_I(t) = v_0 t\,.$$

for the inertial frame. The results are as follows.

((**Mathematica**))

```
Clear["Global`"];

eq1 = D[x[t], {t, 2}] == 2 ω D[y[t], t] + ω^2 x[t];
eq2 = D[y[t], {t, 2}] == -2 ω D[x[t], t] + ω^2 y[t];

eq3 = DSolve[{eq1, eq2, x[0] == 0, y[0] == 0, x'[0] == 0,  y'[0] == v0},
    {x[t], y[t]}, t] // FullSimplify;

x1[t_] = x[t] /. eq3[[1]]

t v0 Sin[t ω]

y1[t_] = y[t] /. eq3[[1]]

t v0 Cos[t ω]
```



Fig. Path of particle projected radially outward from center as viewed from inertial frame of reference.



Fig. Path of particles projected radially outward from center as viewed rotating frame of reference. The rotation direction is counter clockwise.



Fig. Path of particles projected radially outward from center as viewed rotating frame of reference. The rotation direction is clockwise.

12. Another method: solving differential equation using complex number

Now we define the complex number

$$u_R = x_R + iy_R$$

The differential equation:

$$\begin{aligned}\ddot{u}_R &= \ddot{x}_R + i\ddot{y}_R \\ &= 2\omega\dot{y}_R + \omega^2 x_R + i(-2\omega\dot{x}_R + \omega^2 y_R) \\ &= \omega^2(x_R + iy_R) - 2i\omega(\dot{x}_R + i\dot{y}_R) \\ &= \omega^2 u_R - 2i\omega\dot{u}_R\end{aligned}$$

with the initial condition

$$u_R(0) = 0, \qquad \dot{u}_R(0) = iv_0$$

The solution is as follows.

$$u_R(t) = iv_0 t e^{-i\omega t}$$

$$x(t) = \frac{1}{2}[u(t) + u^*(t)] = v_0 t \sin(\omega t)$$

$$y(t) = \frac{1}{2i}[u(t) - u^*(t)] = v_0 t \cos(\omega t)$$

((**Mathematica**))

```
Clear["Global`"]; Clear[u];
exp_* := exp /. {Complex[re_, im_] :> Complex[re, -im]};
eq1 = D[u[t], {t, 2}] == -2 ω ⅈ D[u[t], t] + ω^2 u[t];
eq2 = DSolve[{eq1, u[0] == 0, u'[0] == ⅈ v0}, u[t], t] // FullSimplify;
u1[t_] = u[t] /. eq2[[1]]
```

$i\, e^{-i t \omega}\, t\, v0$

```
x1[t_] = Simplify[ExpToTrig[1/2 (u1[t] + u1[t]*)]]
```

$t\, v0\, \mathrm{Sin}[t\, \omega]$

```
y1[t_] = Simplify[ExpToTrig[(u1[t] - u1[t]*)/(2 ⅈ)]]
```

$t\, v0\, \mathrm{Cos}[t\, \omega]$

---

13. The 3D motion in the presence of only the Coriolis force



$$\boldsymbol{F}_{cor} = 2m\boldsymbol{v}_R \times \boldsymbol{\Omega}$$

$\boldsymbol{F}_{cor}$ acts as a centripetal force, giving rise to the uniform circular motion. The circular motion occurs in a plane perpendicular to $\boldsymbol{\Omega}$ (along the z_I axis).

$$m\frac{{v_R}^2}{r_R} = 2mv_R\Omega_0$$

or

$$\frac{r_R}{v_R} = \frac{1}{2\Omega_0}$$

The period T is

$$T = \frac{2\pi r_R}{v_R} = \frac{\pi}{\Omega_0}$$

We solve the differential equation using Mathematica

$$\dot{x}_R = 2\Omega_0 y_R \sin(\beta)$$

$$\dot{y}_R = -2\Omega_0 x_R \sin(\beta) - 2\Omega_0 z_R \cos(\beta) + v_0$$

$$\dot{z}_R = 2\Omega_0 y_R \cos(\beta)$$

with the initial conditions;

$$\dot{x}_R(t=0) = 0\,, \quad \dot{y}_R(t=0) = v_0\,, \qquad \dot{z}_R(t=0) = 0$$

$$x_R(t=0) = y_R(t=0) = z_R(t=0) = 0$$

The solutions are as follows.

$$x_R(t) = \frac{v_0}{\Omega_0}\sin\beta \sin^2(\Omega_0 t)\,,$$

$$y_R(t) = \frac{v_0}{2\Omega_0}\sin(2\Omega_0 t)$$

$$z_R(t) = \frac{v_0}{\Omega_0}\cos\beta \sin^2(\Omega_0 t)\,.$$

We make a plot of the motion of the system using the ParametricPlot3D . As is expected, we have the uniform circular motion on a plane perpendicular to the z_I axis.



14. Mathematica

The differential equation

```
Clear["Global`*"];
eq1 = D[x[t], t] == 2 Ω0 y[t] Sin[β];
eq2 = D[y[t], t] ==
   -2 Ω0 Sin[β] x[t] - 2 Ω0 Cos[β] z[t] + v0;
eq3 = D[z[t], t] == 2 Ω0 Cos[β] y[t];
eq4 = {x[0] == 0, y[0] == 0, z[0] == 0};
seq1 =
  DSolve[Join[{eq1, eq2, eq3}, eq4 ],
    {x[t], y[t], z[t]}, t] // FullSimplify;
```

```
x[t_] = x[t] /. seq1[[1]]
```

$$\frac{v0\,\mathrm{Sin}[\beta]\,\mathrm{Sin}[t\,\Omega 0]^2}{\Omega 0}$$

```
y[t_] = y[t] /. seq1[[1]]
```

$$\frac{v0\,\mathrm{Sin}[2\,t\,\Omega 0]}{2\,\Omega 0}$$

```
z[t_] = z[t] /. seq1[[1]]
```

$$\frac{v0\,\mathrm{Cos}[\beta]\,\mathrm{Sin}[t\,\Omega 0]^2}{\Omega 0}$$

```
β = 40 °; v0 = 1; Ω0 = 10;
f1 = ParametricPlot3D[100 {x[t], y[t], z[t]},
  {t, 0, 10}, Boxed → False, Axes → None,
  PlotStyle → {Blue, Thick}];
f2 = Graphics3D[
  {Red, Thick, Arrow[{{0, 0, 0}, {4, 0, 0}}],
   Arrow[{{0, 0, 0}, {0, 4, 0}}],
   Arrow[{{0, 0, 0}, {0, 0, 4}}],
   Arrow[{{0, 0, 0}, {-4 Cos[β], 0, 4 Sin[β]}}],
   Text[Style["xR", Black, 12], {4.3, 0, 0}],
   Text[Style["yR", Black, 12], {0, 4.3, 0}],
   Text[Style["zR", Black, 12], {0, 0, 4.3}],
   Text[Style["zI", Black, 12],
    {-4.3 Cos[β], 0, 4.3 Sin[β]}]}];
Show[f1, f2, PlotRange → All]
```

15. Analogy between the Lorentz force and Coriolis force

The analogy between the Coriolis force and the Lorentz force is as follows. Suppose that the Lorentz force is defined by

$$\boldsymbol{F}_{Lor} = \frac{q}{c}\boldsymbol{v}_R \times \boldsymbol{B}$$

where q is the charge and c is the velocity of light. When the effective magnetic field **B** is related to the angular velocity $\boldsymbol{\Omega}$ as

$$\boldsymbol{\Omega} = \frac{q\boldsymbol{B}}{2mc}$$

The direction of the angular velocity is parallel or antiparallel to that of the effective magnetic field. Then we have

$$\boldsymbol{F}_{Lor} = \frac{q}{c}\boldsymbol{v}_R \times \frac{2mc}{q}\boldsymbol{\Omega} = 2m(\boldsymbol{v}_R \times \boldsymbol{\Omega}) = \boldsymbol{F}_{cor}$$



Fig. Configuration of Lorentz force, where $\boldsymbol{\Omega} = \dfrac{q\boldsymbol{B}}{2mc}$.

---

16. Coriolis force and centrifugal force

$$\boldsymbol{F}_{cor} + \boldsymbol{F}_{centri} = 2m\boldsymbol{v}_R \times \boldsymbol{\Omega} - m[\boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_R)]$$

with

$$\boldsymbol{\Omega} = \Omega_0(-\sin\lambda, 0, \cos\lambda) = \Omega_0(-\cos\beta, 0, \sin\beta)$$

where β is the latitude,

$$\beta = \frac{\pi}{2} - \lambda .$$

The equation of the motion is given by

$$\ddot{x}_R = \Omega_0 \sin\beta(2\dot{y}_R + \Omega_0 z_R \cos\beta + \Omega_0 x_R \sin\beta)$$

$$\ddot{y}_R = \Omega_0(-2\dot{x}_R \sin\beta + \Omega_0 y_R - 2\dot{z}_R \cos\beta)$$

$$\ddot{z}_R = \Omega_0 \cos\beta(2\dot{y}_R + \Omega_0 z_R \cos\beta + \Omega_0 x_R \sin\beta)$$

We use the initial conditions given by

$$x_R(t=0) = x_0, \qquad y_R(t=0) = 0, \qquad z_R(t=0) = 0$$

and

$$\dot{x}_R(t=0) = 0, \qquad \dot{y}_R(t=0) = v_0, \qquad \dot{z}_R(t=0) = 0$$

---

17. The in-plane (2D) motion in the presence of only the Coriolis force

Suppose that $z(t) = 0$.

The equation of the motion is given by

$$\ddot{x}_R = \Omega_0 \sin\beta(2\dot{y}_R)$$

$$\ddot{y}_R = \Omega_0(-2\dot{x}_R \sin\beta)$$

or

$$\dot{x}_R = 2\Omega_0 \sin\beta(y_R)$$

$$\dot{y}_R = -2\Omega_0 \sin\beta(x_R) + v_0$$

with the initial conditions;

$$x_R(t=0) = 0, \quad y_R(t=0) = 0, \quad \dot{x}_R(t=0) = 0, \quad \dot{y}_R(t=0) = 0$$

The solution of the above differential equation is given by

$$x_R(t) = \frac{v_0}{\Omega_0 \sin\beta}\sin^2(\Omega_0 \sin\beta t) = \frac{v_0}{2\Omega_0 \sin\beta}[1-\cos(2\Omega_0 \sin\beta t)]$$

$$y_R(t) = \frac{v_0}{2\Omega_0 \sin\beta}\sin(2\Omega_0 \sin\beta t)$$

These points are located on a circle given by

$$(x_R(t) - \frac{v_0}{2\Omega_0 \sin\beta})^2 + [y_R(t)]^2 = \left(\frac{v_0}{2\Omega_0 \sin\beta}\right)^2$$

in the $\{x_R, y_R)$ plane. The center of the circle is $(\frac{v_0}{2\Omega_0 \sin\beta}, 0)$. The radius is $\frac{v_0}{2\Omega_0 \sin\beta}$.

18. Mathematica

We assume that $v_0 = 1$, $\Omega_0 = 10$, β is changed as a parameter; β = 20, 30, 40°, 50°, 60°, 70°, 80°, and 90°. (the clock-wise rotation)

```
Clear["Global`*"];
eq1 = D[x[t], t] == 2 Ω0 y[t] Sin[β];
eq2 = D[y[t], t] == -2 Ω0 Sin[β] x[t] + v0;
eq3 = {x[0] == 0, y[0] == 0};
seq1 =
 DSolve[Join[{eq1, eq2}, eq3 ], {x[t], y[t]},
   t] // FullSimplify;

x[t_] = x[t] /. seq1[[1]]
```

$$\frac{v0\,Csc[\beta]\,Sin[t\,\Omega0\,Sin[\beta]]^2}{\Omega0}$$

```
y[t_] = y[t] /. seq1[[1]]
```

$$\frac{v0\,Csc[\beta]\,Sin[2\,t\,\Omega0\,Sin[\beta]]}{2\,\Omega0}$$

```
v0 = 1; Ω0 = 10;
f1 = ParametricPlot[
  Evaluate[Table[100 {x[t], y[t]},
    {β, 20 °, 90 °, 10 °}]], {t, 0, 10},
  Axes → None, AspectRatio → Automatic,
  PlotStyle → Table[{Hue[0.1 i], Thick},
    {i, 0, 10}]];
f2 =
 Graphics[{Black, Thin,
   Arrow[{{0, 0}, {32, 0}}],
   Arrow[{{0, -15}, {0, 15}}],
   Text[Style["xR", Black, 12], {33, 0}],
   Text[Style["yR", Black, 12], {0, 16}],
   Text[Style["β=30°", Black, 12], {28, 0}],
   Text[Style["β=60°", Black, 12], {14, 0}],
   Text[Style["β=90°", Black, 12], {9, 0}]}];
Show[f1, f2, PlotRange → All]
```



19. Combination of Coriolis and centrifugal forces (2D plane)

Suppose that $z_R(t) = 0$. We only consider the planar motion in the (x_R, y_R) plane

The equation of the motion:

$$\ddot{x}_R = \Omega_0 \sin\beta(2\dot{y}_R + \Omega_0 x_R \sin\beta)$$

$$\ddot{y}_R = \Omega_0(-2\dot{x}_R \sin\beta + \Omega_0 y_R)$$

We use the initial conditions given by

$$x_R(t=0) = x_0, \qquad y_R(t=0) = 0.$$

and

$$\dot{x}_R(t=0) = 0, \qquad \dot{y}_R(t=0) = v_0,$$

We solve the differential equation numerically by using the Mathematica.



Fig. The in-plane motion in the (x_R, y_R) plane under the influence of the Coriolis force and centrifugal force. The latitude $\beta = 40°$.

20. Mathematica

```
Clear["Global`*"]; Ω0 = 1; β = 40 °;
Coriolis[v0_, tmax_] :=
 Module[{numsol, numgraph},
  numsol =
   NDSolve[
    {D[x[t], {t, 2}] == Ω0 Sin[β] (2 y'[t] + x [t] Ω0 Sin[β]),
     D[y[t], {t, 2}] == Ω0 (y[t] Ω0 - 2 x'[t] Sin[β]),
     x[0] == 0, y[0] == 0, x'[0] == 0, y'[0] == v0},
    {x[t], y[t]}, {t, 0, tmax}] ;
  numgraph = ParametricPlot[
    Evaluate[{x[t], y[t]} /. numsol[[1]]],
    {t, 0, tmax}, PlotStyle → {Hue[v0], Thick}]];

g1 = Table[Coriolis[v0, 6], {v0, 0.1, 1.0, 0.1}];
g2 =
 Graphics[
  {Text[Style["v0=0.1", Black, 12], {-2, -1.5}],
   Text[Style["v0=0.2", Black, 12], {-2, -3}],
   Text[Style["v0=0.3", Black, 12], {-2, -4.5}],
   Text[Style["v0=0.4", Black, 12], {-2, -5.5}],
   Text[Style["v0=0.5", Black, 12], {-2, -6.5}],
   Text[Style["v0=0.6", Black, 12], {-2, -7.8}],
   Text[Style["v0=0.7", Black, 12], {-2, -8.8}],
   Text[Style["v0=0.8", Black, 12], {-2, -10}],
   Text[Style["v0=0.9", Black, 12], {-2, -11}],
   Text[Style["v0=1.0", Black, 12], {-2, -12}],
   Text[Style["β=40°", Black, 12], {-7, -1.0}],
   Text[Style["xR", Black, 12], {8, 1}],
   Text[Style["yR", Black, 12], {1, 2}],
   Text[Style["β=40°", Black, 12], {-7, -1.0}] }];
Show[g1, g2, PlotRange → All]
```

21. Free fall and the Coriolis force



z_I
Ω_0
Ω_0
z_R
λ
y_R
r_R
λ
x_R
β
y_I
ϕ
x_I



$$m\boldsymbol{a} = \boldsymbol{F}_{gravity} + \boldsymbol{F}_{Corioli}$$

with

where

$$\boldsymbol{F}_{gravity} = (0,0,-mg)$$

$$\boldsymbol{\Omega} = (-\Omega_0 \sin\lambda, 0, \Omega_0 \cos\lambda)$$

$$\begin{aligned}\boldsymbol{F}_{Corioli} &= -2m(\boldsymbol{\Omega}\times\boldsymbol{v})\\ &= (2m\Omega_0\cos(\lambda)\dot{y}, -2m\Omega_0[\cos(\lambda)\dot{x}+\sin(\lambda)\dot{z}], 2m\Omega_0\sin(\lambda)\dot{y})\end{aligned}$$

Equation of motion:

$$\ddot{x} = 2\Omega_0\cos(\lambda)\dot{y}$$

$$\ddot{y} = -2\Omega_0[\cos(\lambda)\dot{x}+\sin(\lambda)\dot{z}]$$

$$\ddot{z} = -g + 2\Omega_0 \sin(\lambda)\dot{y}$$

Since, $\lambda = \dfrac{\pi}{2} - \beta$ we have

$$\ddot{x} = 2\Omega_0 \sin(\beta)\dot{y}$$

$$\ddot{y} = -2\Omega_0[\sin(\beta)\dot{x} + \cos(\beta)\dot{z}]$$

$$\ddot{z} = -g + 2\Omega_0 \cos(\beta)\dot{y}$$

We use the initial condition: the initial velocity: $(u_0,0,0)$, and the initial position: $(0,0,z_0)$.

$$\dot{x} = 2\Omega_0 y \sin(\beta)$$

$$\dot{y} = -2\Omega_0 x \sin(\beta) - 2\Omega_0(z - z_0)\cos(\beta)$$

$$\dot{z} = -gt + 2\Omega_0 \cos(\beta) y$$

The exact solution of these differential is as follows.

$$x = \frac{g}{8\Omega_0{}^2}[-1 + 2\Omega_0{}^2 t^2 + \cos(2\Omega_0 t)]\sin(2\beta)$$

$$y = \frac{g}{4\Omega_0{}^2}[2\Omega_0 t - \sin(2\Omega_0 t)]\cos(\beta)$$

$$z = \frac{g}{8\Omega_0{}^2}[\{-1 - 2\Omega_0{}^2 t^2 + \frac{8z_0\Omega_0{}^2}{g} + \cos(2\Omega_0 t)\} + \cos(2\beta)\{-1 + 2\Omega_0{}^2 t^2 + \cos(2\Omega_0 t)\}]$$

In the limit of $\Omega_0 t \to 0$, we have

$$x = \frac{1}{6} g t^2 \sin(\beta)\cos(\beta)(\Omega_0 t)^2$$

$$y = \frac{1}{3} g t^2 (\Omega_0 t) \cos(\beta)$$

$$z = z_0 - \frac{1}{2} g t^2 [1 - \frac{1}{3} (\Omega_0 t)^2 \cos^2(\beta)]$$

The remarkable thing about this solution is that a freely falling object doies not fall straight down. Instead the Coriolis force causes it to curve slightly to the positive y direction.

Suppose that we use the approximation;

$$x = 0,$$

$$y = \frac{1}{3} g t^2 (\Omega_0 t) \cos(\beta),$$

$$z = z_0 - \frac{1}{2} g t,$$

to the order of $\Omega_0 t$.

When $z = 0$, (falling from the height $z_0 = h$ to the ground $z = 0$),

$$t = \sqrt{\frac{2h}{g}}.$$

Then the value of y is obtained as

$$y = \frac{1}{3} g \Omega_0 (\frac{2h}{g})^{3/2} \cos(\beta)$$

At Binghamton, NY, the latitude is $\beta = 42.0986$ (N). For $h = 100$ m, the value of y is evaluated as

$y = 1.625$ cm.

where

$$\Omega_0 = \frac{2\pi}{24 \times 3600} = 7.27221 \times 10^{-5} \text{ rad/s}$$

It is a small deflection, but certainly detectable.

22. Mathematica

```
Clear["Global`*"]
```

$$\Omega 0 = \frac{2\,\pi}{24 \times 3600} \ \text{// N;}$$

```
rule1 = {h → 100, β → 42.0986 °, g → 9.8};
```

$$y1 = \frac{1}{3}\, g\, \Omega 0 \left(\frac{2\,h}{g}\right)^{3/2} \text{Cos}[\beta] \ \text{/. rule1 // N}$$

```
0.0162509
```

23. Mathematica

The differential equations (Se. 11) can be solved using the Mathematica.

```
Clear["Global`*"];
eq1 = D[x[t], t] == 2 Ω0 y[t] Sin[β];
eq2 = D[y[t], t] ==
   -2 Ω0 Sin[β] x[t] - 2 Ω0 Cos[β] (z[t] - z0);
eq3 = D[z[t], t] == -g t + 2 Ω0 Cos[β] y[t];
eq4 = {x[0] == 0, y[0] == 0, z[0] == z0,
   x'[0] == 0, y'[0] == 0, z'[0] == 0};
seq1 =
  DSolve[Join[{eq1, eq2, eq3}, eq4],
    {x[t], y[t], z[t]}, t] // FullSimplify;
```

```
x[t_] = x[t] /. seq1[[1]]
```

$$\frac{g\left(-1+2\,t^2\,\Omega 0^2+\mathrm{Cos}[2\,t\,\Omega 0]\right)\mathrm{Sin}[2\,\beta]}{8\,\Omega 0^2}$$

```
y[t_] = y[t] /. seq1[[1]]
```

$$-\frac{g\,\mathrm{Cos}[\beta]\,(-2\,t\,\Omega 0+\mathrm{Sin}[2\,t\,\Omega 0])}{4\,\Omega 0^2}$$

```
z[t_] = z[t] /. seq1[[1]]
```

$$\frac{1}{8\,\Omega 0^2}\left(-g-2\,g\,t^2\,\Omega 0^2+8\,z0\,\Omega 0^2+g\,\mathrm{Cos}[2\,t\,\Omega 0]+g\,\mathrm{Cos}[2\,\beta]\left(-1+2\,t^2\,\Omega 0^2+\mathrm{Cos}[2\,t\,\Omega 0]\right)\right)$$

```
Series[x[t], {t, 0, 4}] // Normal // Simplify
```

$$\frac{1}{6}\,g\,t^4\,\Omega 0^2\,\mathrm{Cos}[\beta]\,\mathrm{Sin}[\beta]$$

```
Series[y[t], {t, 0, 4}] // Normal // Simplify
```

$$\frac{1}{3}\,g\,t^3\,\Omega 0\,\mathrm{Cos}[\beta]$$

```
Series[z[t], {t, 0, 4}] // Normal // Simplify
```

$$-\frac{g\,t^2}{2}+z0+\frac{1}{6}\,g\,t^4\,\Omega 0^2\,\mathrm{Cos}[\beta]^2$$

24. Foucault pendulum



Fig. The latitude $\beta = 90° - \lambda$.

Assume the Earth is a sphere rotating about the z_I axis with constant angular velocity ω. Choose a co-ordinate system on Earth with the ***k*** axis along the vertical, the ***xR*** axis pointing South, and the ***yR*** axis pointing East. β is a latitude of the observer. λ is the colatitude. $\lambda = 90° - \beta$.

The pendulum bob is acted upon by two real forces: the tension in the string (*T*) and gravity (*mg*). It is also acted upon by two fictitious forces: Coriolis force and the centrifugal force. For convenience we neglect the centrifugal force.

$$m\boldsymbol{a} = \boldsymbol{T} + \boldsymbol{F}_{gravity} + \boldsymbol{F}_{Corioli}$$

with

$$\boldsymbol{T} = T\boldsymbol{n} = \frac{T}{L}(-x, -y, L - z)$$

where

$$\boldsymbol{n} = \frac{1}{L}(0 - x, 0 - y, L - z)$$

$$\boldsymbol{F}_{gravity} = (0, 0, -mg)$$

$$\boldsymbol{\Omega} = (-\Omega_0 \sin\lambda, 0, \Omega_0 \cos\lambda)$$

$$\begin{aligned}\boldsymbol{F}_{Corioli} &= -2m(\boldsymbol{\Omega} \times \boldsymbol{v}) \\ &= (2m\Omega_0 \cos(\lambda)\dot{y}, -2m\Omega_0[\cos(\lambda)\dot{x} + \sin(\lambda)\dot{z}], 2m\Omega_0 \sin(\lambda)\dot{y})\end{aligned}$$

Equation of motion:

$$\ddot{x}_R = -\frac{T}{mL}x_R + 2\Omega_0 \cos(\lambda)\dot{y}_R$$

$$\ddot{y}_R = -\frac{T}{mL}y_R - 2\Omega_0[\cos(\lambda)\dot{x}_R + \sin(\lambda)\dot{z}_R]$$

$$\ddot{z}_R = -g + \frac{T}{mL}(L - z_R) + 2\Omega_0 \sin(\lambda)\dot{y}_R$$

The pendulum is very long so that the string is essentially vertical at all times.

$$T = mg$$

$$\ddot{x}_R = -\frac{g}{L}x_R + 2K\dot{y}_R$$

$$\ddot{y} \approx -\frac{g}{L} y - 2K\dot{x}$$

where

$$K = \Omega_0 \cos(\lambda) = \Omega_0 \cos(\frac{\pi}{2} - \beta) = \Omega_0 \sin\beta \,,$$

and

$$\frac{g}{L} = \omega^2 \,.$$

where β is the latitude of the location on the Earth. Then we have the differential equations,

$$\ddot{x}_R = -\omega^2 x_R + 2K\dot{y}_R$$

$$\ddot{y}_R = -\omega^2 y_R - 2K\dot{x}_R$$

We define the complex number as

$$u_R = x_R + iy_R$$

$$\ddot{u}_R = \ddot{x}_R + i\ddot{y}_R = -\omega^2 (x_R + iy_R) - 2iK(\dot{x}_R + i\dot{y}_R) = -\omega^2 u_R - 2iK\dot{u}_R$$

The diffrential equation is then given by

$$\ddot{u}_R + 2iK\dot{u}_R + +\omega^2 u_R = 0$$

with the initial condition

$$u_R(t=0) = x_R(t=0) + iy_R(t=0) = 0 \,,\; \dot{u}_R(t=0) = \dot{x}_R(t=0) + i\dot{y}_R(t=0) = v_0 \,.$$

The solution of the differential equation is obtained by using the Mathematica. The final result is as follows.

$$x_R(t) = \frac{v_0 \cos(Kt)\sin(\Omega_1 t)}{\Omega_1}$$

$$y_R(t) = -\frac{v_0 \sin(Kt)\sin(\Omega_1 t)}{\Omega_1}$$

where

$$\Omega_1 = \sqrt{\omega^2 + K^2}$$

Since

$$\Omega_1 \approx \omega$$

we get

$$x(t) \approx \frac{v_0 \sin(\omega t)}{\omega}\cos(Kt)$$

$$y(t) = -\frac{v_0 \sin(\omega t)}{\omega}\sin(Kt)$$

with

$$K = \Omega_0 \sin\beta$$



Fig. The oscillation of the Foucault pendulum. $K = 1$. $\omega = 3$. $v_0 = 1$. $t = 1$ - 15.

25. Mathematica

```
Clear["Global`*"];

exp_* := exp /. {Complex[re_, im_] :> Complex[re, -im]};

eq1 = u''[t] == -2 ⅈ K u'[t] - ω^2 u[t];

eq2 = Simplify[DSolve[{eq1, u[0] == 0, u'[0] == v0}, u[t], t]];

srule1 = {Sqrt[-(ω^2 + K^2)] -> ⅈ Ω1, 1/Sqrt[-(ω^2 + K^2)] -> -ⅈ/Ω1};

u1[t_] = FullSimplify[u[t] /.eq2[[1]] /.srule1]
```

$$\frac{e^{-i K t}\, v0\, \mathrm{Sin}[t\, \Omega 1]}{\Omega 1}$$

```
x1[t_] = Simplify[ExpToTrig[1/2 (u1[t] + u1[t]*)]]
```

$$\frac{v0\, \mathrm{Cos}[K\, t]\, \mathrm{Sin}[t\, \Omega 1]}{\Omega 1}$$

```
y1[t_] = Simplify[ExpToTrig[(u1[t] - u1[t]*)/(2 ⅈ)]]
```

$$-\frac{v0\, \mathrm{Sin}[K\, t]\, \mathrm{Sin}[t\, \Omega 1]}{\Omega 1}$$

```
x11[t_] = x1[t] /. {Ω1 → √(K² + ω²)};

y11[t_] = y1[t] /. {Ω1 → √(K² + ω²)};

ParametricPlot[Evaluate[{x11[t], y11[t]} /. {K → 1, ω → 3, v0 → 1 }],
 {t, 0, 100}, PlotStyle → {Red, Thick}, Background → LightGray]
```



26. Foucault's pendulum, Pantheon, Paris

The first public exhibition of a Foucault pendulum took place in February 1851 in the Meridian of the Paris Observatory. A few weeks later Foucault made his most famous pendulum when he suspended a 28 kg brass-coated lead bob with a 67 meter long wire from the dome of the Panthéon, Paris. The plane of the pendulum's swing rotated clockwise 11° per hour, making a full circle in 32.7 hours. The original bob used in 1851 at the Panthéon was moved in 1855 to the Conservatoire des Arts et Métiers in Paris. A second temporary installation was made for the 50th anniversary in 1902.
http://en.wikipedia.org/wiki/Foucault_pendulum

The angular velocity of the Earth:

$$\Omega_0 = \frac{2\pi}{24 \times 3600} = 7.27221 \times 10^{-5} \text{ rad/s}$$

Latitude of Panthéon, Paris, France

$\beta = 48.8742°$ N. $\lambda = 90° - \beta = 41.1258°$.

The detail of the Foucault pendulum:

$L = 67$ m. $m = 47$ kg.

The angular velocity of the pendulum

$$\omega = \sqrt{\frac{g}{L}} = \sqrt{\frac{9.80}{67}} = 0.382451 \text{ rad/s}$$

$$T_0 = \frac{2\pi}{\omega} = 16.4287 \text{ s.}$$

$$K = \Omega_0 \sin\beta = 0.753267 \text{ x } 7.27221 \times 10^{-5} = 5.477916 \text{ x } 10^{-5} \text{ rad/s}$$

$$\Omega_1 = \sqrt{\omega^2 + K^2} = 0.382451 \text{ rad/s} \approx \omega.$$

The period:

$$T_K = \frac{2\pi}{K} = 114700 \text{ s} = 31 \text{ hours } 51 \text{ min } 40 \text{ sec}$$

Fig. Foucault's pendulum, Pantheon, Paris
http://en.wikipedia.org/wiki/Foucault_pendulum

We note that

$$T_K = \frac{24}{\sin\beta} \text{ hours}$$

depends on the latitude. When $\beta = 90°$ (North Pole), T_K is exactly equal to 24 hours.

27. Foucault's pendulum in New York City (United Nations)



Fig. Foucault pendulum which hangs in the United Nations Building in New York City.

The sphere takes approximately 36 hours and 45 minutes to complete its cycle. The time of rotation has been found mathematically to vary in accordance with the latitude of the location of the pendulum. At the North Pole, where the pendulum would be directly above the earth's axis and the latitude is 90 degrees, the time of rotation is 24 hours. At the equator, where the latitude is 0 degree, the plane of the pendulum would not shift at all. At other latitudes the Foucault effect varies, becoming more pronounced nearer the poles.

Latitude of New York City, NY (U.S.A.):

$\beta = 40.7142°$ N. $\lambda = 90° - \beta = 49.2858°$.

$L = 23$ m. $m = 95$ kg.

$$\omega = \sqrt{\frac{g}{L}} = \sqrt{\frac{9.80}{23}} = 0.652753 \text{ rad/s}$$

$$T_0 = \frac{2\pi}{\omega} = 9.62567 \text{ s.}$$

$$K = \Omega_0 \sin\beta = 4.74356 \text{ x } 10^{-5} \text{ rad/s}$$

$$\Omega_1 = \sqrt{\omega^2 + K^2} = 0.652753 \text{ rad/s} \approx \omega.$$

$$T_K = \frac{2\pi}{K} = 132457 \text{ s} = 36 \text{ hours } 47 \text{ min } 37 \text{ sec}$$

29. Meteological phenomena due to the Coriolis force

The Coriolis force plays a significant role in many meteological phenomena. Mases of air tend to move from a region of high pressure to a region of low pressure, the so-called pressure gradient The wind continues parallel to the isobars, circulating in the Northern Hemisphere in a counter clock-wise direction around the center of low pressure, and in a clock-wise direction around the center of high pressure, when viewed from the above.



Fig. Air flows near the site with low pressure in the northern Hemisphere. Near the center of low pressure, the air flow rotates in counter clock-wise.



Fig. Air flows near the site with high pressure in the northern Hemisphere. Near the center of high pressure, the air flow rotates in clock-wise.

REFERENCES

H. Goldstein, C. Poole, and J. Safko, Classical Mechanics, third edition (John Wesley & Sons, 2000).

C. Kittel, Mechanics, Berkley physics Course vol.1 second edition (McGraw Hill, 1973)

G.L. Baker and J.A. Blackburn, The Pendulum A Case Study in Physics (Oxford University Press, 2005)

V. Bargar and M. Olsson, Classical Mechanics A Modern Perspective(McGraw-Hill, 1973).

P. Hamil, Intermediate Dynamics (Jones and Bartlett Publishers, Boston, 2010).

J.R. Taylor, Classical Mechanics (University Science Books, 2005).

J.M. Finn, Classical Mechanics (Infinity Science Press, 2008).

W. Greiner, Classical Mechanics, second edition (Springer, 2010).

S.T. Thornton and J.B. Marion, Classical Dynamics of Particles and Systems, 5-th edition (Brook's/Cole - Thomson Learning, 2004)

R.D. Gregory, Classical Mechanics An Undergraduate Text (Cambridge University Press, 2006).

APPENDIX-I List of the Foucault pendulum in the world

Location	Length (m)	Mass (kg)
Pantheon, Paris	67	47 (1995 bob)
UN building, NYC	23	90
Franklin Institute, Philadelphia	26	410
Univ. of Guelph, Canada	0.83	4.5
Univ. of Maryland	14	28
Yale University	11	12
Griffith Observatory, Los Angeles	12	100
Ryerson Library, Michigan	22	14
Ottawa High School, Michigan	9	14
Univ. of Wisconsin, Madison	14.5	30
Morrison Planeterium	9	87
Science Museum, London	20	9

APPENDIX-II Method using the Lagrange equation

The Lagrangian L is given by

$$L = T - U = \frac{1}{2}m[(\dot{x}_I)^2 + (\dot{y}_I)^2 + (\dot{z}_I)^2 - U(x_I, y_I, z_I)$$

where T is the kinetic energy and U is the potential energy. Using the relation

$$\begin{pmatrix} x_I \\ y_I \\ z_I \end{pmatrix} = \begin{pmatrix} \cos(\omega t) & -\sin(\omega t) & 0 \\ \sin(\omega t) & \cos(\omega t) & 0 \\ 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} x_R \\ y_R \\ z_R \end{pmatrix}$$

the new Lagrangian L can be rewrittem in terms of the new variables, x_R, y_R, z_R as

$$L(x_R, y_R, z_R, \dot{x}_R, \dot{y}_R, \dot{z}_R = \frac{1}{2} m[\omega^2 (x_R{}^2 + y_R{}^2) + 2\omega(x_R \dot{y}_R - \dot{x}_R y_R) + (\dot{x}_R)^2 + (\dot{y}_R)^2 + (\dot{z}_R)^2] - U(x_R, y_R, z_R)$$

since

$$U(x_R, y_R, z_R) = U(x_I, y_I, z_I)$$

The Lagrange equaion

$$\frac{d}{dt}\left(\frac{\partial L}{\partial \dot{x}_R}\right) = \frac{\partial L}{\partial x_R}, \qquad \frac{d}{dt}\left(\frac{\partial L}{\partial \dot{y}_R}\right) = \frac{\partial L}{\partial y_R}, \qquad \frac{d}{dt}\left(\frac{\partial L}{\partial \dot{z}_R}\right) = \frac{\partial L}{\partial z_R}$$

$$m\ddot{x}_R = m\omega^2 x_R + 2m\omega\dot{y}_R - \frac{\partial U(x_R, y_R, z_R)}{\partial x_R}$$

$$m\ddot{y}_R = m\omega^2 y_R - 2m\omega\dot{x}_R - \frac{\partial U(x_R, y_R, z_R)}{\partial y_R}$$

$$m\ddot{z}_R = -\frac{\partial U(x_R, y_R, z_R)}{\partial z_R}$$

We have the form of vectors

$$m\boldsymbol{a}_R = -2m(\boldsymbol{\Omega} \times \boldsymbol{v}_R) - m[\boldsymbol{\Omega} \times (\boldsymbol{\Omega} \times \boldsymbol{r}_R)] - Grad[U]$$

APPENDIX-III

colatitude (λ) The latitude measured down from the North Pole (instead of up from the equator, as is more usual with geographers).

latitude (β) It is geographic coordinate that specifies the north-south position of a point on the Earth's surface. Latitude is an angle (defined below) which ranges from 0° at the equator to 90° (North or South) at the poles.
